// This Pine Script™ code is subject to the terms of the Mozilla Public License 2.0 at https://mozilla.org/MPL/2.0/
// © VanHe1sing

//@version=6
indicator("Lowess Channel + (RSI) [ChartPrime]", overlay = true, max_labels_count = 500)


// --------------------------------------------------------------------------------------------------------------------}
// 𝙐𝙎𝙀𝙍 𝙄𝙉𝙋𝙐𝙏𝙎
// --------------------------------------------------------------------------------------------------------------------{
length      = input.int(25, group = "Lowess Channel")
multi       = input.float(2, "Bands", step = 0.1, group = "Lowess Channel")
rsi_displ   = input.bool(true, "RSI", inline = "rsi", group = "RSI")
rsi_len     = input.int(14, "", inline = "rsi", group = "RSI")
bandwidth   = 10
col_up      = input.color(color.lime, "", group = "Color", inline = "color")
col_dn      = input.color(color.purple, "", group = "Color", inline = "color")
col_bnds    = input.color(#3d9140, "", group = "Color", inline = "color")

// --------------------------------------------------------------------------------------------------------------------}
// 𝙄𝙉𝘿𝙄𝘾𝘼𝙏𝙊𝙍 𝘾𝘼𝙇𝘾𝙐𝙇𝘼𝙏𝙄𝙊𝙉𝙎
// --------------------------------------------------------------------------------------------------------------------{
lowess_smooth(src, length, bandwidth) =>
    sum_weights = 0.0
    sum_weighted_y = 0.0
    sum_weighted_xy = 0.0
    sum_weighted_x2 = 0.0
    sum_weighted_x = 0.0

    for i = 0 to length - 1
        x = float(i)
        weight = math.exp(-0.5 * (x / bandwidth) * (x / bandwidth))
        y = nz(src[i], 0)

        sum_weights := sum_weights + weight
        sum_weighted_x := sum_weighted_x + weight * x
        sum_weighted_y := sum_weighted_y + weight * y
        sum_weighted_xy := sum_weighted_xy + weight * x * y
        sum_weighted_x2 := sum_weighted_x2 + weight * x * x

    mean_x = sum_weighted_x / sum_weights
    mean_y = sum_weighted_y / sum_weights

    beta = (sum_weighted_xy - mean_x * mean_y * sum_weights) / (sum_weighted_x2 - mean_x * mean_x * sum_weights)
    alpha = mean_y - beta * mean_x

    alpha + beta * float(length / 2)  // Centered smoothing

get_col(src, len)=>
    var col = color(na)

    if ta.rising(src,len)
        col := col_up

    if ta.falling(src, len)
        col := col_dn

    col

atr = ta.atr(200)
rsi = float(na)
if rsi_displ
    rsi := ta.rsi(close, rsi_len)

band(src, multi, side = true, display = true)=>
    lowess_smooth(src+(atr*multi*(side?1:-1)), length, bandwidth)

h2  =   band(close, multi*2)
h1  =   band(close, multi)
m   =   band(close, 0)
l1  =   band(close, multi, false)
l2  =   band(close, multi*2, false)

dashed_col = bar_index % 2 == 0 ? col_bnds : color(na)

color_trend = get_col(m, 3) 


// --------------------------------------------------------------------------------------------------------------------}
// 𝙑𝙄𝙎𝙐𝘼𝙇𝙄𝙕𝘼𝙏𝙄𝙊𝙉
// --------------------------------------------------------------------------------------------------------------------{
if color_trend != color_trend[1]
    label.new(bar_index-1, m[1], "◆", color = color(na), style = label.style_label_center, textcolor = color_trend, size = size.large)

if barstate.islast and rsi_displ
    hl = (h2-l2) / 100
    rsi_vl = l2 + hl*rsi

    m_ = line.new(bar_index+4, l2 + hl *50, bar_index+6, l2 + hl *50, color = color_trend)
    L = line.new(bar_index+5, h2, bar_index+5, l2, color = color_trend, style = line.style_arrow_both)
    L1 = label.new(bar_index+5, rsi_vl, "◈", color = color(na), style = label.style_label_center, textcolor = color_trend, size = size.large)
    L2 = label.new(bar_index+5, rsi_vl, str.tostring(rsi, "◂ ##.#"), color = color(na), style = label.style_label_left, textcolor = chart.fg_color, size = size.normal)
    L3 = label.new(bar_index+5, h2, "RSI", color = color(na), style = label.style_label_down, textcolor = chart.fg_color, size = size.normal)

    line.delete(m_[1])
    line.delete(L[1])
    label.delete(L1[1])
    label.delete(L2[1])
    label.delete(L3[1])


plot(h2, "Higher Band 2", color = dashed_col)
plot(h1, "Higher Band 1", color = dashed_col)
plot(m, "Middle", color = get_col(m, 3), linewidth = 2)
plot(l1, "Lower Band 1", color = dashed_col)
plot(l2, "Lower Band 2", color = dashed_col)
// --------------------------------------------------------------------------------------------------------------------}
